Overview
There are three objectives in this lab:

- understand the mathematical reasoning associated with a real-world example,
- learn to define a piecewise-defined function in Maple, and
- learn to set up and solve a system of equations in Maple.

Maple Essentials
Important Maple commands introduced in this lab are:

<table>
<thead>
<tr>
<th>Command</th>
<th>Description</th>
<th>Example</th>
</tr>
</thead>
<tbody>
<tr>
<td>restart</td>
<td>clears the internal memory</td>
<td>should always start a session with restart;</td>
</tr>
<tr>
<td>piecewise</td>
<td>piecewise-defined function</td>
<td>piecewise(x&lt;0, 2*x, x&gt;0 and x&lt;2, x^2);</td>
</tr>
<tr>
<td>assign</td>
<td>assignment</td>
<td>assign({x=3, y=1}); assign(solutions);</td>
</tr>
</tbody>
</table>

The general syntax to define a piecewise-defined function

\[ F(x) = \begin{cases} 
  f_1(x), & \text{cond}_1 \\
  f_2(x), & \text{cond}_2 \\
  \vdots & \vdots \\
  f_n(x), & \text{cond}_n 
\end{cases} \]

in Maple is:

\[ F := x -> \text{piecewise}(\text{cond}_1, f_1(x), \text{cond}_2, f_2(x), \ldots, \text{cond}_n, f_n); \]

where each \( \text{cond}_i \) is an inequality and each \( f_i(x) \) is an expression. Maple does not recognize double inequalities, so if your condition is \( a \leq x < b \) you would write \( x >= a \) and \( x < b \).

Related course material/Preparation
Review properties of the first derivative.

The Problem: Design a Roller Coaster
Suppose we are asked to design a simple ascent and drop roller coaster with an overall horizontal displacement of 200 feet. By studying pictures of our favorite roller coasters, we decide that our roller coaster should begin the ascent along a line \( y = f_1(x) \) of slope 1.5 for the first 20ft horizontally. Next, it should continue the ascent and then begin the drop along a parabola \( y = f_2(x) = ax^2 + bx + c \) for the next 100ft horizontally. Finally, we want it to begin a soft landing at 30ft above the ground along a cubic \( y = f_3(x) = dx^3 + ex^2 + fx + g \) for the last 80ft. The main task is to find \( \{a, b, c, d, e, f, g\} \) that will ensure that the track is smooth at transition points. We also would like to know the maximum height of the roller coaster.

Assignment/Project 1
Your assignment/project 1 is to repeat this lab for a bigger roller coaster and to prepare a neat and complete project report. The due date will be specified by your TA.

Solving the problem using Maple
1. One should always start a Maple session with
   
   ```markdown
   > restart;
   ```

   This clears the internal memory so that Maple acts (almost) as if just started and is very helpful in case that you made a mistake (you likely will) and want to start over.

2. Let us first define our functions. If we choose the origin as our starting point, then it is easy to see \( f_1(x) = 1.5x \) as it is a line of slope 1.5 that passes through \((0,0)\):

   ```markdown
   > f1:=x-> 1.5*x;
   > f2:=x-> a*x^2+b*x+c;
   > f3:=x-> d*x^3+e*x^2+f*x+g;
   ```

3. Since our roller coaster consists of three curves, it can be set up mathematically as a piecewise-defined function:

   
   \[
   F(x) = \begin{cases} 
   f_1(x), & 0 \leq x \leq 20 \\
   f_2(x), & 20 < x < 120 \\
   f_3(x), & 120 \leq x \leq 200 
   \end{cases}
   \]

   Let us define \( F \) as a function in Maple as follows:
   
   ```markdown
   > F:= x -> piecewise(x<=20,f1(x), x>20 and x<120,f2(x), x>=120 and x<=200,f3(x));
   ```

   **Note:** You can verify your function by typing \( F(x) \);

4. We will also need the first derivatives of our functions. (If you do not see why, you will soon.) To find and assign the derivatives, right-click over the function and choose differentiate. Then right-click over the derivative function and choose assign to a name. Name the derivatives \( df_1 \), \( df_2 \), \( df_3 \), and \( dF \), respectively.

5. Obviously, we want \( F(x) \) to be continuous (so our passengers do not perish). This means that our functions should be equal at transition points. So we get the following equations:

   ```markdown
   > eq1:=f1(20)=f2(20);
   > eq2:=f2(120)=f3(120);
   ```

6. If we are to have a smooth track, we cannot have abrupt changes in direction, so the first derivative \( F'(x) \) should also be continuous. That is, the first derivatives of our functions should also be equal at transition points. So we get:

   ```markdown
   > eq3:=df1(20)=df2(20);
   > eq4:=df2(120)=df3(120);
   ```

7. To start our landing at 30ft above the ground for the last 80ft, we would have:

   ```markdown
   > eq5:=f3(120)=30;
   ```

8. Finally, in order to have a soft landing, the track should be tangent to the ground at the end:

   ```markdown
   > eq6:=f3(200)=0;
   > eq7:=df3(200)=0;
   ```

9. We now have a system of 7 equations and 7 unknowns. We solve using the solve command and assign the solutions as follows:

   ```markdown
   > values:=solve({eq1,eq2,eq3,eq4,eq5,eq6,eq7} , {a,b,c,d,e,f,g});
   ```

10. Plug those solved values into functions once for all:

    ```markdown
    > assign(values);
    ```

11. Let us see what our coaster looks like with the following plot command (notice that we want the same scale for both x and y):

    ```markdown
    > plot(F(x), x=0..200, y=-50..150);
    ```

12. To find the maximum height, find where the graph has a horizontal tangent line and evaluate \( F(x) \) at each of those x-values. The largest is the maximum height of the coaster.

    ```markdown
    > solve(dF(x)=0, x);
    ```

   **Note:** We know that \( F'(x) = 0 \) when we have a horizontal tangent line (a slope of 0). This occurs at both local maximums and minimums. More detailed discussions will be given in Chapter 5.